
            

Todo lo que necesitamos es un microcon-
trolador Atmega 32 y un circuito integrado 
Ethernet serie ENC28J60. Este circuito inte-
grado es una gran solución para otras fami-
lias de microcontroladores, tales como PIC, 
dsPIC etc. El conector RJ-45 CJCBA8HF1Y0 de 
CviLux se usa para la conexión a la red Ether-
net. Un diodo LED conectado al PORTA.0 del 
microcontrolador, simulan una aplicación do-
méstica que quiere el control. 

El compilador mikroBASIC for AVR contie-
ne la librería SPI_Ethernet que simplifi cará 
considerablemente el proceso de escritura 
de un programa para el microcontrolador. 
Usando unas pocas rutinas de esta libre-
ría, es posible crear el programa que nos 
permitirá controlar aplicaciones eléctricas 
en nuestra casa a través de un explorador 
web. 

Para ello, es necesario realizar las siguientes 
operaciones dentro del programa:
 
Paso 1.  Crear una página html a través  
 de la cual arrancar el microcontro 
 lador. Importar el código como  
 un bloque de texto (“string”). 

Paso 7.  Leer los datos recibidos en un bu 
 cle sin fi n. 

La parte más importante del programa es la 
función Spi_Ethernet_userTCP, que procesa 
todos los comandos recibidos. 
Después de recibir la petición “GET” del na-
vegador web, enviada desde nuestro orde-
nador a la dirección IP del sistema de control, 
el microcontrolador responderá con una pá-
gina web almacenada en su memoria. Esta 
página será mostrada automáticamente en 
la pantalla del ordenador por el navegador 
web. Cuando se recibe el comando ON, el 
diodo LED conectado a PORTA.0 se encen-
derá.
Del mismo modo, cuando se recibe el coman-
do OFF, el diodo LED se apaga. Si en lugar de un 
diodo LED tenemos un relé, es posible controlar 
cualquier aplicación como una lámpara, un siste-
ma de seguridad, un sistema de calefacción, etc. 

Automatización doméstica, con-
trol doméstico, casa inteligente o digi-

tal, son sólo diferentes nombres para confort, 
conveniencia, seguridad y ahorro de energía. Los 

sistemas de ahorro de energía están adquiriendo, hoy día, 
una mayor importancia. Incluso si pensamos que dichos sistemas 

son muy caros, podemos asegurar que también son bastante baratos si 
los fabricamos por nosotros mismos. Hay varias formas de controlar un sistema 

doméstico inteligente. Uno de ellos es a través de Ethernet.

Paso 2.  Confi gurar las direcciones IP, DNS,  
 Gateway y máscaras de Subred  
 proporcionadas por nuestro   
 proveedor de Internet. 

Por ejemplo, nuestros parámetros locales 
de red son los siguientes:

IP: 192.168.20.60 (dirección del Sistema de Control)
DNS: 192.168.20.1 (dirección del Domain Name Sys 
             tem o Sistema de Nombres de Dominio)
GATEWAY: 192.168.20.6 (dirección de la pasare 
     la o Gateway)
SUBNET: 255.255.255.0 (máscara de Subred)

Paso 3.  Deshabilitar las entradas analógi 
 cas de PORTA. El terminal del mi 
 crocontrolador debe ser borrado  
 y confi gurado como una salida.

Paso 4.  Inicializar el módulo SPI del mi- 
 crocontrolador Atmega 32. 

Paso 5.  Inicializar el módulo Serie Ether- 
 net del circuito integrado   
 ENC28J60. 

Paso 6.  Escribir el código dentro de 
 la función Spi_Ethernet_userTCP  
 que, después de recibir el co 
 mando a través del explorador  
 web, encenderá/apagará el dio 
 do LED conectado al PORTA.0 

program enc_ethernet

dim myMacAddr   as byte[6]     ‘ my MAC address
    myIpAddr    as byte[4]     ‘ my IP address
    gwIpAddr    as byte[4]     ‘ gateway (router) IP address
    ipMask      as byte[4]     ‘ network mask (for example : 255.255.255.0)
    dnsIpAddr   as byte[4]     ‘ DNS server IP address
    indexPage   as string[523] ‘ default html page
    getRequest  as byte[20]    ‘ HTTP request buff er
    httpHeader  as string[30]          ‘ HTTP header
    httpMimeTypeHTML as string[13]     ‘ HTML MIME type
    httpMimeTypeScript as string[14]   ‘ TEXT MIME type

‘ mE ehternet NIC pinout
  SPI_Ethernet_Rst as sbit at PORTB.B0
  SPI_Ethernet_CS  as sbit  at PORTB.B1
  SPI_Ethernet_Rst_Direction as sbit at DDRB.B0
  SPI_Ethernet_CS_Direction  as sbit at DDRB.B1
‘ end ethernet NIC defi nitions

sub function putString(dim s as ^byte) as word
  result = 0
  while(s^ <> 0)
    Spi_Ethernet_putByte(s^)
    Inc(s)
    Inc(result)
  wend
end sub

sub function SPI_Ethernet_UserTCP(dim byref remoteHost as byte[4],
                                  dim remotePort, localPort, reqLength as word) as word
dim  cnt as word            ‘ my reply length
     tmp as string[10]

  if(localPort <> 80) then  ‘ I listen only to web request on port 80
    result = 0 exit
  end if

  ‘ get 15 fi rst bytes only of the request, the rest does not matter here
  for cnt = 0 to 14  getRequest[cnt] = SPI_Ethernet_getByte() next cnt
  getRequest[cnt] = 0

  tmp = “GET /”
  if(memcmp(@getRequest, @tmp, 5) <> 0) then  ‘ only GET method
    result = 0  exit
  end if

  tmp = “ON”
  if(memcmp(@getRequest+11, @tmp, 2) = 0) then    ‘ do we have ON command
    PORTA.B0 = 1                                 ‘ set PORTA bit0
  else
    tmp = “OFF”
    if(memcmp(@getRequest+11, @tmp, 3) = 0) then ‘ do we have OFF command
      PORTA.B0 = 0                               ‘ clear PORTA bit0
    end if
  end if

  if (PORTA.B0) then
    tmp = “#FFFF00” memcpy(@indexPage+340, @tmp, 6) ‘ highlight (yellow) ON
    tmp = “#4974E2” memcpy(@indexPage+431, @tmp, 6) ‘ clear OFF
  else
    tmp = “#4974E2” memcpy(@indexPage+340, @tmp, 6) ‘ clear ON
    tmp = “#FFFF00” memcpy(@indexPage+431, @tmp, 6) ‘ highlight (yellow) OFF
  end if
  cnt = putString(@httpHeader)              ‘ HTTP header
  cnt = cnt + putString(@httpMimeTypeHTML)  ‘ with HTML MIME type
  cnt = cnt + putString(@indexPage)         ‘ HTML page fi rst part
  result = cnt   ‘ return to the library with the number of bytes to transmit
end sub

sub function SPI_Ethernet_UserUDP(dim byref remoteHost as byte[4],
                                  dim remotePort, destPort, reqLength as word) as word
  result = 0 ‘ back to the library with the length of the UDP reply
end sub

main:
  ‘ set PORTA as output
  PORTA0_bit = 0 ‘ clear PORTA.B0
  DDRA.B0    = 1 ‘ set PORTA.B0 as output (rele control pin)
  
  httpHeader         = “HTTP/1.1 200 OK”+chr(10)+”Content-type: “
  httpMimeTypeHTML   = “text/html”+chr(10)+chr(10)
  httpMimeTypeScript = “text/plain”+chr(10)+chr(10)
  indexPage =
  “<html><head><title>mikroElektronika</title></head><body>”+
  “<h3 align=center>MikroElektronika Home Automatization System</h3>”+
  “<form name=”+Chr(34)+”input”+Chr(34)+” action=”+Chr(34)+”/”+Chr(34)+” method=”+
  Chr(34)+”get”+Chr(34)+”><table align=center width=200 bgcolor=#4974E2 border=2><tr>”+
  “<td align=center colspan=2><font size=4 color=white><b>Heat Control</b></font>”+
  “</td></tr><tr><td align=center bgcolor=#4974E2><input name=”+Chr(34)+”tst1”+
  Chr(34)+” width=60 type=”+Chr(34)+”submit”+Chr(34)+” value=”+Chr(34)+”ON”+
  Chr(34)+”></td><td align=center bgcolor=#FFFF00><input name=”+Chr(34)+”tst2”+
  Chr(34)+” type=”+Chr(34)+”submit”+Chr(34)+” value=”+Chr(34)+”OFF”+Chr(34)+
  “></td></tr></table></form></body></html>”

  myMacAddr[0] = 0x00 myMacAddr[1] = 0x14 myMacAddr[2] = 0xA5
  myMacAddr[3] = 0x76 myMacAddr[4] = 0x19 myMacAddr[5] = 0x3F
  ipMask[0]    = 255 ipMask[1]    = 255 ipMask[2]    = 255 ipMask[3]    =  0
  myIpAddr[0]  = 192 myIpAddr[1]  = 168 myIpAddr[2]  =  20 myIpAddr[3]  = 60
  gwIpAddr[0]  = 192 gwIpAddr[1]  = 168 gwIpAddr[2]  =  20 gwIpAddr[3]  =  6
  dnsIpAddr[0] = 192 dnsIpAddr[1] = 168 dnsIpAddr[2] =  20 dnsIpAddr[3] =  1

  ‘ starts ENC28J60 with: reset bit on PORTB.F0, CS bit on PORTB.F1,
  ‘                       my MAC & IP address,  full duplex
  SPI1_Init_Advanced(_SPI_MASTER, _SPI_FCY_DIV4, _SPI_CLK_LO_LEADING)
  SPI_Rd_Ptr = @SPI1_Read
  SPI_Ethernet_UserTCP_Ptr = @SPI_Ethernet_UserTCP
  SPI_Ethernet_UserUDP_Ptr = @SPI_Ethernet_UserUDP
  SPI_Ethernet_Init(myMacAddr, myIpAddr, SPI_Ethernet_FULLDUPLEX)

  ‘ dhcp will not be used here, so use preconfi gured addresses
  SPI_Ethernet_confNetwork(ipMask, gwIpAddr, dnsIpAddr)

  while true  ‘ do forever
      SPI_Ethernet_doPacket() ‘ process incoming Ethernet packets
  wend
end.

Ejemplo 1:  Programa que demuestra el control a través de Ethernet

... making it simple                                                               

El código para este ejemplo ha sido escrito para microcontroladores PIC® en 
lenguaje C, Basic y Pascal, del mismo modo que los programas han sido escritos para 
microcontroladores dsPIC® y AVR®. Todo ello lo pueden encontrar en nuestra página 
web:  www.mikroe.com/en/article/N

O
TA

:

El control de cualquier aplicación doméstica consiste en la introducción de la 
dirección IP del sistema de control en el navegador web y especifi car los co-
mandos deseados. 
Por supuesto, es posible controlar más de un terminal del microcontrolador, lo 
cual nos permite gobernar un gran número de aplicaciones o un sistema de 
automatización doméstico completo.

La captura de pantalla ilustra la 
página web mostrada por el na-
vegador después de introducir la 
dirección IP del sistema de control. 
En nuestro ejemplo, al pulsar sobre 
los botones ON y OFF provocare-
mos que el diodo LED se encienda 
y se apague, simulando el control 
de un sistema de calefacción.

Mas abajo está la lista de las funciones, ya creadas, contenidas en la librería SPI 
Ethernet Library. Esta librería esta integrada en el compilador mikroBASIC for AVR. 

 * Funciones de la librería SPI Ethernet Library usadas en el programa

Spi_Ethernet_Init()*            Inicia el controlador ENC28J60
Spi_Ethernet_Enable()                  Habilita de tráfi co de la red 
Spi_Ethernet_Disable()              Deshabilitar el tráfi co de la red 
Spi_Ethernet_doPacket()*         Paquetes de procesos recibido 
Spi_Ethernet_putByte()             Almacena un byte  
Spi_Ethernet_putBytes()                 Almacena bytes 
Spi_Ethernet_putConstBytes()  Almacena bytes como constantes
Spi_Ethernet_putString()*               Almacena string 
Spi_Ethernet_putConstString()*                                Almacena string como constante 
Spi_Ethernet_getByte()*                                     Obtener un byte
Spi_Ethernet_getBytes()                                         Obtener bytes 
Spi_Ethernet_UserTCP()*   Código de manejo TCP
Spi_Ethernet_UserUDP()    Código de manejo UDP 
Spi_Ethernet_getIpAddress()           Obtiene dirección IP 
Spi_Ethernet_getGwIpAddress()         Obtiene dirección Gateway o  pasarela
Spi_Ethernet_getDnsIpAddress()    Obtiene dirección DNS 
Spi_Ethernet_getIpMask()              Obtiene máscara de dirección IP 
Spi_Ethernet_confNetwork()*                        Establece los parámetros de red 
Spi_Ethernet_arpResolve()   Envía una petición ARP 
Spi_Ethernet_sendUDP()     Envía un paquete UDP 
Spi_Ethernet_dnsResolve()   Envía una petición DNS 
Spi_Ethernet_initDHCP()                        Envía una petición DHCP
Spi_Ethernet_doDHCPLeaseTime()          Tiempo de proceso
Spi_Ethernet_renewDHCP()   Petición de renovar DHCP 

Spi_Init()  Inicializar el módulo SPI del microcontrolador
memcpy()  Copia posiciones de la memoria RAM del microcontrolador
memcmp()  Compara posiciones de la memoria RAM del microcontrolador

Otras funciones mikroBASIC for PIC 
usadas en el programa:

Esquema eléctrico 1. Conexión del módulo Serie Ethernet al Atmega 32

